Solución de la Evaluación

Ejercicio 2

a) `Xs = [2, 4, 6, 8]`

La primitiva findall encuentra todas las soluciones del objetivo (append(Q,[X],TL), T=..TL, call(T)). Este último agrega X al final de Q y transforma el resultado en un término para ser evaluado.

Entonces, la consulta todos_q([par_menor, 9], Xs) devuelve una lista con las soluciones de par_menor(9, X).

Croquis del árbol de derivación:



b) Xs = []

Falla en la primera rama y se corta la búsqueda.

Croquis del árbol de derivación:



c) Xs = [2,4,6,8]

El objetivo between(1,9,X) falla cuando M>9. Por lo tanto, se encuentran todas las soluciones.

Croquis del árbol de derivación:



d) Xs = [2]

El objetivo (append(Q,[X],TL), T=..TL, call(T),!) deriva en (par_menor(9,X), !).

Cuando se encuentra la primer solución de par_menor(9,X), el cut corta la búsqueda.



Ejercicio 3

a)

```
gram(P, M) -->
        gram(a, 0, P),
        {P > 0},
        gram(b, 0, M),
        {M > 0},
        {PM is P*M},
        gram(c, 0, PM).

gram(_, N, N) -->
        [].

gram(S, N0, N) -->
        [S],
        {N1 is N0+1},
        gram(S, N1, N).
```

b)

```
todos_falsos([]).
todos_falsos([G|Gs]) :-
        not(soluciones([G])),
        todos_falsos(Gs).

soluciones([], Nots) :-
        todos_falsos(Nots).

soluciones([not(G)|Gs], Nots) :-
        ground(G),
        not(soluciones(G)),
        soluciones(Gs, Nots).

soluciones([not(G)|Gs], Nots) :-
        not(ground(G)),
        soluciones(Gs, [G|Nots]).

soluciones([G|Gs], Nots) :-
        not(G = not(_)),
        regla(G, B),
        append(B, Gs, Gs1),
        soluciones(Gs1, Nots).

soluciones(Goal) :-
        soluciones(Goal, []).

regla(H, B) :-
        clause(H, B0),
        regla1(B0, B).

regla1(true, []).
regla1((A,B), [A|C]) :-
        regla1(B, C).
regla1(A, [A]) :-
        not(A = (_,_)),
        not(A = true).
```

Ejercicio 4

```
kreinas(K, N, Reinas) :-
    noatacan(K, 1, N, [], Reinas),
    not((between_(1, N, F)
        ,between_(1, N, C)
        ,not((member_(R, Reinas)
             ,ataca((F, C), R))))).

noatacan(0, _, _, Reinas, Reinas).
noatacan(K, F0, N, Reinas0, Reinas) :-
    K > 0,
    between_(1, N, C),
    between_(F0, N, F),
    not((member_(R, Reinas0)
        ,ataca(R, (F, C)))),
    F1 is F+1,
    K1 is K-1,
    noatacan(K1, F1, N, [(F, C)|Reinas0], Reinas).

ataca((F, _), (F, _)).
ataca((_, C), (_, C)).
ataca((F1, C1), (F2, C2)) :-
    F is abs(F1-F2),
    C is abs(C1-C2),
    0 is F-C.

between_(N, M, N) :-
    N =< M.
between_(N, M, P) :-
    N < M,
    N1 is N+1,
    between_(N1, M, P).

member_(X, [X|_]).
member_(X, [_|Xs]) :-
    member_(X, Xs).
```

Ejercicio 5

a) Basta con encontrar una refutación SLD:

$\leftarrow p(X,Y)$

1

$\leftarrow \underline{q(X_1,f(a),Z_1)}, r(Y_1)$

2

$\leftarrow r(Y_1)$

3

$\square$

$\{X/a, Y/X_3\}$

1	$p(X_1,Y_1) \leftarrow q(X_1,f(a),Z_1), r(Y_1)$ $\theta_1 = \{X/X_1, Y/Y_1\}$
2	$q(X_2, f(X_2), b)$ $\theta_2 = \{X_1/a, X_2/a, Z_2/b\}$
3	$r(X_3)$ $\theta_3 = \{Y_1/X_3\}$

b) θ es una respuesta correcta: se puede obtener a partir de la respuesta computada $\{X/a, Y/X_3\}$ componiéndole la la sustitución $\{X_3/b\}$. Sin embargo, θ no es una respuesta computada. Sólo hay una única cláusula para p(X,Y), con lo

que toda refutación SLD genera un objetivo como el planteado en (a). El único literal donde interviene Y es r(Y), y este último literal se elimina con la última regla, que no instancia argumentos.

c) Es posible, por ejemplo, si el intérprete utiliza una regla de computación en donde se selecciona al átomo de más a la derecha. En ese caso, el árbol obtenido para ←q(X,Y,a) es finitamente fallado (ver parte d).

En cambio, si el intérprete utiliza una regla de computación que selecciona al átomo de más a la izquierda, no se puede probar el objetivo, pues el árbol SLD tiene una rama infinita.



1	$q(X_1,Y_1,Z_1) \leftarrow q(f(X_1),g(Y_1),h(Z_1)),s(X_1)$ $\theta_1 = \{X/X_1, Y/Y_1, Z_1/a\}$
2	$q(X_2,Y_2,Z_2) \leftarrow q(f(X_2),g(Y_2),h(Z_2)), s(X_2)$ $\theta_2 = \{X_2/f(X_1), Y_2/g(Y_1), Z_2/h(a)\}$

d)



falla

1	$q(X_1,Y_1,Z_1) \leftarrow q(f(X_1),g(Y_1),h(Z_1)), s(X_1)$ $\theta_1 = \{X/X_1, Y/Y_1, Z_1/a\}$